
NAG Library Routine Document

D01GBF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

D01GBF returns an approximation to the integral of a function over a hyper-rectangular region, using a
Monte–Carlo method. An approximate relative error estimate is also returned. This routine is suitable for
low accuracy work.

2 Specification

SUBROUTINE D01GBF (NDIM, A, B, MINCLS, MAXCLS, FUNCTN, EPS, ACC, LENWRK,
WRKSTR, FINEST, IFAIL)

&

INTEGER NDIM, MINCLS, MAXCLS, LENWRK, IFAIL

REAL (KIND=nag_wp) A(NDIM), B(NDIM), FUNCTN, EPS, ACC, WRKSTR(LENWRK),
FINEST

&

EXTERNAL FUNCTN

3 Description

D01GBF uses an adaptive Monte–Carlo method based on the algorithm described in Lautrup (1971). It is
implemented for integrals of the form:

Z b1

a1

Z b2

a2

� � �
Z bn

an

f x1; x2; . . . ; xnð Þ dxn � � � dx2dx1.

Upon entry, unless LENWRK has been set to the minimum value 10� NDIM, the routine subdivides the
integration region into a number of equal volume subregions. Inside each subregion the integral and the
variance are estimated by means of pseudorandom sampling. All contributions are added together to
produce an estimate for the whole integral and total variance. The variance along each coordinate axis is
determined and the routine uses this information to increase the density and change the widths of the sub-
intervals along each axis, so as to reduce the total variance. The total number of subregions is then
increased by a factor of two and the program recycles for another iteration. The program stops when a
desired accuracy has been reached or too many integral evaluations are needed for the next cycle.

4 References

Lautrup B (1971) An adaptive multi-dimensional integration procedure Proc. 2nd Coll. Advanced Methods
in Theoretical Physics, Marseille

5 Parameters

1: NDIM – INTEGER Input

On entry: n, the number of dimensions of the integral.

Constraint: NDIM � 1.

2: AðNDIMÞ – REAL (KIND=nag_wp) array Input

On entry: the lower limits of integration, ai, for i ¼ 1; 2; . . . ; n.

3: BðNDIMÞ – REAL (KIND=nag_wp) array Input

On entry: the upper limits of integration, bi, for i ¼ 1; 2; . . . ; n.

4: MINCLS – INTEGER Input/Output

On entry: must be set

either to the minimum number of integrand evaluations to be allowed, in which case
MINCLS � 0;

or to a negative value. In this case, the routine assumes that a previous call had been made
with the same parameters NDIM, A and B and with either the same integrand (in which case
D01GBF continues calculation) or a similar integrand (in which case D01GBF begins the
calculation with the subdivision used in the last iteration of the previous call). See also
WRKSTR.

On exit: contains the number of integrand evaluations actually used by D01GBF.

5: MAXCLS – INTEGER Input

On entry: the maximum number of integrand evaluations to be allowed. In the continuation case
this is the number of new integrand evaluations to be allowed. These counts do not include zero
integrand values.

Constraints:

MAXCLS > MINCLS;
MAXCLS � 4� NDIMþ 1ð Þ.

6: FUNCTN – REAL (KIND=nag_wp) FUNCTION, supplied by the user. External Procedure

FUNCTN must return the value of the integrand f at a given point.

The specification of FUNCTN is:

FUNCTION FUNCTN (NDIM, X)

REAL (KIND=nag_wp) FUNCTN

INTEGER NDIM

REAL (KIND=nag_wp) X(NDIM)

1: NDIM – INTEGER Input

On entry: n, the number of dimensions of the integral.

2: XðNDIMÞ – REAL (KIND=nag_wp) array Input

On entry: the coordinates of the point at which the integrand f must be evaluated.

FUNCTN must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D01GBF is called. Parameters denoted as Input must not be changed by
this procedure.

7: EPS – REAL (KIND=nag_wp) Input

On entry: the relative accuracy required.

Constraint: EPS � 0:0.

8: ACC – REAL (KIND=nag_wp) Output

On exit: the estimated relative accuracy of FINEST.

9: LENWRK – INTEGER Input

On entry: the dimension of the array WRKSTR as declared in the (sub)program from which
D01GBF is called.

For maximum efficiency, LENWRK should be about

3� NDIM� MAXCLS=4ð Þ1=NDIM þ 7� NDIM.

If LENWRK is given the value 10� NDIM then the subroutine uses only one iteration of a crude
Monte–Carlo method with MAXCLS sample points.

Constraint: LENWRK � 10� NDIM.

10: WRKSTRðLENWRKÞ – REAL (KIND=nag_wp) array Input/Output

On entry: if MINCLS < 0, WRKSTR must be unchanged from the previous call of D01GBF –
except that for a new integrand WRKSTRðLENWRKÞ must be set to 0:0. See also MINCLS.

On exit: contains information about the current sub-interval structure which could be used in later
calls of D01GBF. In particular, WRKSTRðjÞ gives the number of sub-intervals used along the jth
coordinate axis.

11: FINEST – REAL (KIND=nag_wp) Input/Output

On entry: must be unchanged from a previous call to D01GBF.

On exit: the best estimate obtained for the integral.

12: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, because for this routine the values of the output parameters
may be useful even if IFAIL 6¼ 0 on exit, the recommended value is �1. When the value �1 or 1
is used it is essential to test the value of IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, NDIM < 1,
or MINCLS � MAXCLS,
or LENWRK < 10� NDIM,
or MAXCLS < 4� NDIMþ 1ð Þ,
or EPS < 0:0.

IFAIL ¼ 2

MAXCLS was too small for D01GBF to obtain the required relative accuracy EPS. In this case
D01GBF returns a value of FINEST with estimated relative error ACC, but ACC will be greater
than EPS. This error exit may be taken before MAXCLS nonzero integrand evaluations have
actually occurred, if the routine calculates that the current estimates could not be improved before
MAXCLS was exceeded.

7 Accuracy

A relative error estimate is output through the parameter ACC. The confidence factor is set so that the
actual error should be less than ACC 90% of the time. If you want a higher confidence level then a
smaller value of EPS should be used.

8 Further Comments

The running time for D01GBF will usually be dominated by the time used to evaluate FUNCTN, so the
maximum time that could be used is approximately proportional to MAXCLS.

For some integrands, particularly those that are poorly behaved in a small part of the integration region,
D01GBF may terminate with a value of ACC which is significantly smaller than the actual relative error.
This should be suspected if the returned value of MINCLS is small relative to the expected difficulty of the
integral. Where this occurs, D01GBF should be called again, but with a higher entry value of MINCLS
(e.g., twice the returned value) and the results compared with those from the previous call.

The exact values of FINEST and ACC on return will depend (within statistical limits) on the sequence of
random numbers generated within D01GBF by calls to G05SAF. Separate runs will produce identical
answers.

9 Example

This example calculates the integral
Z 1

0

Z 1

0

Z 1

0

Z 1

0

4x1x3 exp 2x1x3ð Þ
1þ x2 þ x4ð Þ2

dx1dx2dx3dx4 ¼ 0:575364.

9.1 Program Text

! D01GBF Example Program Text
! Mark 24 Release. NAG Copyright 2012.

Module d01gbfe_mod

! D01GBF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: lenwrk = 500, maxcls = 20000, &

ndim = 4, nout = 6
Contains

Function functn(ndim,x)

! .. Function Return Value ..
Real (Kind=nag_wp) :: functn

! .. Scalar Arguments ..
Integer, Intent (In) :: ndim

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: x(ndim)

! .. Intrinsic Procedures ..
Intrinsic :: exp

! .. Executable Statements ..
functn = 4.0E0_nag_wp*x(1)*x(3)**2*exp(2.0E0_nag_wp*x(1)*x(3))/ &

(1.0E0_nag_wp+x(2)+x(4))**2

Return

End Function functn
End Module d01gbfe_mod
Program d01gbfe

! D01GBF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d01gbf, nag_wp
Use d01gbfe_mod, Only: functn, lenwrk, maxcls, ndim, nout

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: acc, eps, finest
Integer :: ifail, mincls

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:), b(:), wrkstr(:)

! .. Executable Statements ..
Write (nout,*) ’D01GBF Example Program Results’

Allocate (a(ndim),b(ndim),wrkstr(lenwrk))

a(1:ndim) = 0.0E0_nag_wp
b(1:ndim) = 1.0E0_nag_wp
eps = 0.01E0_nag_wp
mincls = 1000

ifail = -1
Call d01gbf(ndim,a,b,mincls,maxcls,functn,eps,acc,lenwrk,wrkstr,finest, &

ifail)

Select Case (ifail)
Case (0,2)

Write (nout,*)
Write (nout,99998) ’Requested accuracy = ’, eps
Write (nout,99997) ’Estimated value = ’, finest
Write (nout,99998) ’Estimated accuracy = ’, acc
Write (nout,99999) ’Number of evaluations = ’, mincls

End Select

99999 Format (1X,A,I5)
99998 Format (1X,A,E13.2)
99997 Format (1X,A,F13.5)

End Program d01gbfe

9.2 Program Data

None.

9.3 Program Results

D01GBF Example Program Results

Requested accuracy = 0.10E-01
Estimated value = 0.57572
Estimated accuracy = 0.92E-02
Number of evaluations = 1728

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

D01 – Quadrature D01GBF

Mark 24 D01GBF.5 (last)


	D01GBF
	1 Purpose
	2 Specification
	3 Description
	4 References
	5 Parameters
	NDIM
	A
	B
	MINCLS
	MAXCLS
	FUNCTN
	NDIM in subprogram FUNCTN
	X in subprogram FUNCTN

	EPS
	ACC
	LENWRK
	WRKSTR
	FINEST
	IFAIL

	6 Error Indicators and Warnings
	IFAIL=1
	IFAIL=2

	7 Accuracy
	8 Further Comments
	9 Example
	9.1 Program Text
	9.2 Program Data
	9.3 Program Results


	NAG Fortran Library Manual, Mark 24
	Copyright Statement
	Foreword
	Introduction
	Essential Introduction
	NAG Fortran Library specific documentation
	 NAG Fortran Library News

	NAG SMP Library specific documentation
	Introduction to the NAG Library for SMP & Multicore
	 NAG Library for SMP & Multicore News
	Tuned and Enhanced Routines in the 

	Thread Safety
	Routines Withdrawn or Scheduled for Withdrawal
	Advice on Replacement Calls for Withdrawn/Superseded Routines
	Acknowledgements
	Indexes

	Implementation-specific Details for Users
	Chapters of the Library
	A00 - Library Identification
	A00 Chapter Introduction

	A02 - Complex Arithmetic
	A02 Chapter Introduction

	C02 - Zeros of Polynomials
	C02 Chapter Introduction

	C05 - Roots of One or More Transcendental Equations
	C05 Chapter Introduction

	C06 - Summation of Series
	C06 Chapter Introduction

	C09 - Wavelet Transforms
	C09 Chapter Introduction

	D01 - Quadrature
	D01 Chapter Introduction

	D02 - Ordinary Differential Equations
	D02 Chapter Introduction

	D03 - Partial Differential Equations
	D03 Chapter Introduction

	D04 - Numerical Differentiation
	D04 Chapter Introduction

	D05 - Integral Equations
	D05 Chapter Introduction

	D06 - Mesh Generation
	D06 Chapter Introduction

	E01 - Interpolation
	E01 Chapter Introduction

	E02 - Curve and Surface Fitting
	E02 Chapter Introduction

	E04 - Minimizing or Maximizing a Function
	E04 Chapter Introduction

	E05 - Global Optimization of a Function
	E05 Chapter Introduction

	F - Linear Algebra
	F Chapter Introduction

	F01 - Matrix Operations, Including Inversion
	F01 Chapter Introduction

	F02 - Eigenvalues and Eigenvectors
	F02 Chapter Introduction

	F03 - Determinants
	F03 Chapter Introduction

	F04 - Simultaneous Linear Equations
	F04 Chapter Introduction

	F05 - Orthogonalization
	F05 Chapter Introduction

	F06 - Linear Algebra Support Routines
	F06 Chapter Introduction

	F07 - Linear Equations (LAPACK)
	F07 Chapter Introduction

	F08 - Least Squares and Eigenvalue Problems (LAPACK)
	F08 Chapter Introduction

	F11 - Large Scale Linear Systems
	F11 Chapter Introduction

	F12 - Large Scale Eigenproblems
	F12 Chapter Introduction

	F16 - Further Linear Algebra Support Routines
	F16 Chapter Introduction

	G01 - Simple Calculations on Statistical Data
	G01 Chapter Introduction

	G02 - Correlation and Regression Analysis
	G02 Chapter Introduction

	G03 - Multivariate Methods
	G03 Chapter Introduction

	G04 - Analysis of Variance
	G04 Chapter Introduction

	G05 - Random Number Generators
	G05 Chapter Introduction

	G07 - Univariate Estimation
	G07 Chapter Introduction

	G08 - Nonparametric Statistics
	G08 Chapter Introduction

	G10 - Smoothing in Statistics
	G10 Chapter Introduction

	G11 - Contingency Table Analysis
	G11 Chapter Introduction

	G12 - Survival Analysis
	G12 Chapter Introduction

	G13 - Time Series Analysis
	G13 Chapter Introduction

	H - Operations Research
	H Chapter Introduction

	M01 - Sorting and Searching
	M01 Chapter Introduction

	S - Approximations of Special Functions
	S Chapter Introduction

	X01 - Mathematical Constants
	X01 Chapter Introduction

	X02 - Machine Constants
	X02 Chapter Introduction

	X03 - Inner Products
	X03 Chapter Introduction

	X04 - Input/Output Utilities
	X04 Chapter Introduction

	X05 - Date and Time Utilities
	X05 Chapter Introduction

	X07 - IEEE Arithmetic
	X07 Chapter Introduction




